# Maintainer: Sven-Hendrik Haase <svenstaro@gmail.com>
# Contributor: Alexander Suhoverhov <cy at ngs dot ru>
pkgname=xonotic-data
pkgver=0.8.5
pkgrel=1
pkgdesc="A free, fast-paced crossplatform first-person shooter (data files)"
arch=('any')
url="http://xonotic.org"
license=('GPL')
depends=('bash' 'perl')
makedepends=('unzip')
source=("https://dl.xonotic.org/xonotic-${pkgver}.zip")
sha512sums=('4ffc4b73eeb5f580d178a98419d5b44cbff0c56e356a62baa729b5b7a6c3d43b2b425b123428c9b1bf3f4718eaf61bcf5d62914521cc061c7563a253440c807e')

package() {
  cd Xonotic

  # data
  mkdir -p "$pkgdir"/usr/share/xonotic/
  mv data "$pkgdir"/usr/share/xonotic/

  # server stuff
  cp -r server "$pkgdir"/usr/share/xonotic/

  # key
  install -Dm644 key_0.d0pk "$pkgdir"/usr/share/xonotic/key_0.d0pk
}
# vim: ts=2:sw=2 et:
